

Project 1: Goblet Design

Douglas Meade, Ronda Sanders, and Xian Wu
Department of Mathematics

Instructions

The goal of the project is to design the most visually appealing goblet that meets the following criteria:

- the goblet will be molded using a symmetric mold, that is, the goblet must be a solid of revolution;
- the goblet must hold between 12 and 14 in³ of liquid and uses less than 8 in³ of glass;
- the height of the center of mass must be less than 3 times the radius of the foot so it would be reasonably stable;
- thickness of the glass must be at least 0.07 in at its thinnest point.

Your report is due at the beginning of the next lab meeting and should follow the guidelines set forth in the What is a Project Report? on our lab web. In particular, your report should include the following:

- a detailed description of your design.
- a (2-D) plot of the region to be revolved and a (3-D) plot of the goblet
- detailed numerical results showing that the criteria are satisfied.

An Example in Maple:

For your convenience, this example is available in a Maple worksheet from our lab web.

1. Basic Analyses: By studying our favorite goblets at home, we decide that our new goblet should have a foot (base) of 1.8 inches in radius and 0.5 inches in height. We also like to have a twisted stem of 6 inches long and about 0.2 inches in radius. Finally, we want it to have a bowl that gradually extends to a radius of 1 inch in the first 1.5 inches from the stem and then to a radius of 2 inches in the last 1.5 inches. Put it in sideways and our goblet hence extends from $x = 0$ to $x = 9.5$. The outside profile of our goblet can therefore be described by a piecewise function that consists of four functions f_1 , f_2 , f_3 , and f_4 defined over $0 \leq x < 0.5$, $0.5 \leq x < 6.5$, $6.5 \leq x < 8$, and $8 \leq x \leq 9.5$, respectively. Pay close attention as your TA will explain further details.
2. Working with Maple
 - (a) You should always start a Maple session with `restart` command as it clears the internal memory so that Maple acts (almost) as if just started. In general, you need to start over from very beginning after changes are made.


```
> restart;
```
 - (b) Next, load needed packages using `with` command.


```
> with(plots):
> with(Student[Calculus1]):
```
 - (c) For the foot, we want it to start at $(0, 1.8)$ and end at $(0.5, 0.2)$. Two points determine a line and we can hence define $f_1 = 1.8 - 3.2x$. Ask your TA if you don't see this.


```
> f1:=1.8-3.2*x;
```
 - (d) For a plain stem of radius 0.2, we may just set $f_2 = 0.2$. To have a twisted stem, we will add some small waves to it with $asin(b(x - c))$. For this particular goblet, we take $a = 0.05$ as the height of waves, $b = 2\pi$ to have six waves for our stem from $x = 0.5$ to $x = 6.5$, and $c = 0$ so that waves start at $(0.5, 0.2)$. Note that we do have $f_1(0.5) = f_2(0.5) = 0.2$.


```
> f2:=0.2+0.05*sin(2*Pi*x);
```
 - (e) By our design, the first part of the bowl starts at $(6.5, 0.2)$ and it should be concave down. So let's use $f_3 = a\sqrt{x - 6.5} + 0.2$, where a is a constant to be solved later to satisfy the requirement (equation 1) $f_3(8) = 1$ at $x = 8$. Note that we do have $f_2(6.5) = f_3(6.5) = 0.2$.


```
> f3:=a*sqrt(x-6.5)+0.2;
```

- (f) The second part of the bowl should be concave up and let's try $f_4 = bx^2 + cx + d$.
- ```
> f4:=b*x^2+c*x+d;
```
- (g) Of course, we need to set up three more equations for solving  $b$ ,  $c$ , and  $d$  so that  $f_4$  will be smoothly connected to  $f_3$  at  $x = 8$  and also satisfy the required condition at the end  $x = 9.5$ . For this, we also need to find the derivative of  $f_3$  and  $f_4$ .
- ```
> df3:=diff(f3,x);
> df4:=diff(f4,x);
> eq1:=eval(f3,x=8)=1;
> eq2:=eval(f4,x=8)=eval(f3,x=8);
> eq3:=eval(df4,x=8)=eval(df3,x=8);
> eq4:=eval(f4,x=9.5)=2;
```
- (h) Now, let's solve for a , b , c , and d using `solve` command and assign solutions to values.
- ```
> values:=solve({eq1,eq2,eq3,eq4},{a,b,c,d});
```
- (i) We can then plug those solved values into our functions once for all using `assign` command.
- ```
> assign(values);
```
- (j) We are now ready to put all four functions together as a piecewise function and plot it to see the outside profile.
- ```
> F:=piecewise(x<=0.5,f1,x>0.5 and x<=6.5,f2,x>6.5 and x<=8,f3, 8<x,f4);
> plot(F,x=0..9.5,y=0..4,scaling=constrained);
```
- (k) To get the piecewise function for the inside profile, we shift the bowl part of the outside 0.2 inches to the right. This can be done by substituting  $x$  with  $x - 0.2$  in  $f_3$  and  $f_4$ .
- ```
> g3:=subs(x=x-0.2,f3);
> g4:=subs(x=x-0.2,f4);
> G:=piecewise(x<=6.7,0,x>6.7 and x<=8.2,g3,8.2<x,g4);
> plot(G,x=0..9.5,y=0..5,scaling=constrained);
```
- (l) Here is the goblet. Rotate and/or right click the plot to try some visual effect options!
- ```
> VolumeOfRevolution(F,G,x=0..9.5,output=plot,orientation=[0,180], title='Ha!');
```
- (m) We still need to check the criteria. For the center of mass, we can see that it is located on the  $x$ -axis, say, at  $x = CM$ . It is not too hard to derive the following formula for  $CM$ :

$$CM = \frac{\int_0^{9.5} x(F^2 - G^2) dx}{\int_0^{9.5} (F^2 - G^2) dx}.$$

```
> Capacity:=evalf(VolumeOfRevolution(G,0,x=0..9.5,output=integral));
> Glass:=evalf(VolumeOfRevolution(F,G,x=0..9.5,output=integral));
> Difference:=F-G;
> plot([Difference,0.07],x=6.7..9.5);
> Thickness:=eval(Difference,x=8.05);
> CM:=evalf(int(x*(F^2-G^2),x=0..9.5)/int(F^2-G^2,x=0..9.5));
> Ratio:=CM/eval(F,x=0);
```

3. Conclusion: Looks like we have a stable goblet that satisfies the requirement on amount of glass used. However, the goblet does not hold enough and the glass is too thin. The visual appearance of the goblet can also be improved. Now, it is up to you to design a more interesting and better looking goblet that satisfies all the criteria.

### Acknowledgement

The best design from each section and an over-all winner will be selected. We want to thank MapleSoft for providing prizes for winning projects.